

LE PROGRAMME BASIC

```
10 MODE 1:MEMORY 24600:GOSUB 760:DEFINT a:DIM a(400,2)
20 GOTO 900
30 MODE 1:INK 0,0:INK 1,18:INK 2,6:INK 3,1:PAPER 0:PEN 1
40 BORDER 13
50 can=101:no=1
60 WINDOW #1,2,39,24,24:PAPER #1,3:CLS #1:MOVE 14,32:DRAW 626,32,2:DRAW 626,14:DRAW 14,14:DRAW 14,32
70 ORIGIN 0,50,0,640,400,50:CLG 0:MOVE 0,0:DRAW 639,0:DRAW 639,349:DRAW 0,349:DRAW 0,0
80 LOCATE #1,1,1:PRINT #1,"      ND:      C:
X:      Y:      "
90 ND=0:C=0:X=320:Y=174:PLOT X,Y,1:x2=x:y2=y
100 IF NO=1 THEN D$="D" ELSE D$="A"
110 LOCATE #1,1,1:PRINT #1,D$
120 '
130 '*** DESSIN DES OBJETS ***
140 pa=-2:ph=1:LOCATE #1,8,1:PRINT #1,nd:LOCATE #1,17,1:PRINT #1,c:LOCATE #1,24,1:PRINT #1,x:LOCATE #1,31,1:PRINT #1,y
150 IF nd>=100 THEN 360
160 IF nd=can THEN 320
170 IF INKEY(9)=0 THEN L=16 ELSE L=2
180 IF INKEY(0)=0 THEN y2=y+L
190 IF INKEY(1)=0 THEN x2=x+L
200 IF INKEY(2)=0 THEN y2=y-L
210 IF INKEY(8)=0 THEN x2=x-L
220 IF x2>-1 AND x2<640 AND y2>-1 AND y2<350 AND (x<>x2 OR y<>y2) THEN PLOT x,y,c:c=TEST(x2,y2):x=x2:y=y2:PLOT x,y,1:LOCATE #1,17,1:PRINT #1,c:LOCATE #1,24,1:PRINT #1,x:LOCATE #1,31,1:PRINT #1,y:GOTO 150
230 IF INKEY(55)=0 THEN 270
240 IF INKEY(53)=0 THEN 320
250 IF INKEY(50)=0 THEN 380
260 GOTO 150
270 '
280 ' ***** VALIDATION *****
290 SOUND 1,100,50,15:IF ph=1 THEN PLOT x,y,1:pa=pa+2:a(pa,no)=x:a(pa+1,no)=y:ph=2:c=1:GOTO 310
300 IF ph=2 THEN MOVE a(pa,no),a(pa+1,no):DRAW x,y,1:c=1:pa=pa+2:a(pa,no)=x:a(pa+1,no)=y:ph=1:nd=nd+1:LOCATE #1,8,1:PRINT #1,nd
310 FOR k=1 TO 700:NEXT:GOTO 150
320 '
330 ' ***** CHGT DE DESSIN OU FIN *****
340 IF no=1 THEN no=2:can=nd:nd=0:CLG 0:GOTO 60
350 IF NO=2 THEN 570
360 ' ***** + DE 100 LIGNES *****
370 SOUND 1,300,100:SOUND 1,400,75:FOR K=1 TO 200:NEXT:GOTO 340
380 ' ***** CERCLE *****
390 '
400 IF nd+10>can THEN 150
410 SOUND 1,200,50:FOR i=1 TO 2000:NEXT:cex=x:cey=y
420 IF INKEY(8)=0 THEN x2=x-2
430 IF INKEY(1)=0 THEN x2=x+2
440 IF INKEY(55)=0 THEN SOUND 1,200,40,15:GOTO 470
450 PLOT x,y,c:c=TEST(x2,y2):x=x2:y=y2:PLOT x,y,1:LOCATE #1,24,1:PRINT #1,x:LOCATE #1,31,1:PRINT #1,y:LOCATE #1,17,1:PRINT #1,c
460 GOTO 420
470 ' ***** TRACE DU CERCLE *****
480 '
490 r=ABS(x-cex)
500 DEG:FOR an=0 TO 324 STEP 36
510 x1=cex+COS(an)*r:y1=cey+SIN(an)*r:MOVE x1,y1:pa=pa+2:a(pa,no)=x1:a(pa+1,no)=y1
520 x21=cex+COS(an+36)*r
530 x21=cex+COS(an+36)*r:y21=cey+SIN(an+36)*r:pa=pa+2:a(pa,no)=x21:a(pa+1,no)=y21
540 DRAW x21,y21,1
```

```

550 NEXT
560 nd=nd+10:LOCATE #1,8,1:PRINT #1,nd:GOTO 150
570 '
580 ' ***** TRAITEMENT DES DESSINS **
590 CLS
600 LOCATE 10,10:PRINT "TRAITEMENT DES DESSINS"
610 k=-4:v=nd*8:vr=-2
620 com=com+1:k=k+4:adr=25000-v+k
630 vr=vr+2
640 x=a(vr,1):x2=a(vr,2):y=a(vr+1,1):y2=a(vr+1,2)
650 cox=(x2-x)/19:coy=(y2-y)/19:cy=y-coy:cx=x-cox
660 FOR i=1 TO 20
670 cx2=CINT(cx+I*cox):cy2=CINT(cy+i*coy):adr=adr+v:fo=INT(cx2/256):fa=((cx2/256)-fo)*256:POKE adr,INT(fa):POKE adr+1,fo:fo=INT(cy2/256):fa=((cy2/256)-fo)*256:POKE adr+2,INT(fa):POKE adr+3,fo
680 NEXT i
690 IF com<>nd*2 THEN 620
700 com=0
710 GOTO 900
720 ' ***** AFFICHAGE DES DESSINS ***
730 MODE 1
740 CALL &9C40,25000+(nd*160-8),-(nd*8)*vaf
750 CALL &BB06:GOTO 900
760 ' ***** MACHINE *****
770 '
780 DATA 21,a0,61,e5,dd,5e,00,dd,56,01
790 DATA ed,53,a4,9c,DD,5e,02,dd,56,03
800 DATA ed,53,a6,9c,3e,00,cd,de,bb,e1
810 DATA 01,08,00,09,e5,ed,5b,a4,9c,19
820 DATA cd,85,9c,3e,01,cd,de,bb,e1,e5
830 DATA 01,08,00,cd,85,9c,e1,e5,ed,5b
840 DATA a6,9c,a7,ed,52,20,d5,e1,c9,11
850 DATA 60,a0,ed,b0,ed,5b,60,a0,2a,62
860 DATA a0,cd,c0,bb,ed,5b,64,a0,2a,66
870 DATA a0,cd,f6,bb,c9
880 FOR i=0 TO 94:READ a$:POKE &9C40+i,VAL("&" + a$):NEXT i
890 RETURN

```

```

900  ' ***** CHOIX *****
910  '
920  CLS
930  LOCATE 10,3:PRINT "1  Chargement"
940  LOCATE 10,5:PRINT "2  Sauvegarde"
950  LOCATE 10,7:PRINT "3  Affichage"
960  LOCATE 10,9:PRINT "4  Dessin "
970  LOCATE 10,11:PRINT "5  Mode d'affichage"
980  LOCATE 10,13:PRINT "6  Fin"
990  LOCATE 3,17:PRINT "Appuyer sur l' option cho
isie"
1000  t$=INKEY$:IF T$="" THEN 1000
1010  f=VAL(t$):IF f<1 OR f>6 THEN 1000
1020  ON f GOTO 1040,1060,720,30,1080,1030
1030  MODE 1:END
1040  ' ***** CHARGEMENT *****
1050  CLS:INPUT "Nom du dessin ";n$:LOAD n$:nd=PE
EK(24999):GOTO 900
1060  ' ***** SAUVEGARDE ****
1070  POKE 24999,nd:INPUT "Nom du dessin ";n$:SAV
E n$,b,24999,(nd*160)+1:GOTO 900
1080  ' ***** MODE D' AFFICH. *****
1090  MODE 1:LOCATE 1,1:INPUT "Dessins intermedia
ires :";d$
1100  IF d$="o" OR d$="O" THEN vaf=0 ELSE vaf=1
1110  GOTO 900

```

LE PROGRAMME ASSEMBLEUR

\$9C40	21A061	LD HL,\$61A0	adresse de début des données
\$9C43	E5	PUSH HL	
\$9C44	DD5E00	LD E,(IX+\$00)	
\$9C47	DD5601	LD D,(IX+\$01)	adresse départ et arrivée 1er seg
\$9C4A	ED53A49C	LD (\$9CA4),DE	
\$9C4E	DD5E02	LD E,(IX+\$02)	
\$9C51	DD5603	LD D,(IX+\$03)	
\$9C54	ED53A69C	LD (\$9CA6),DE	
\$9C58	3E00	LD A,\$00	début ss-programme lecture pts
\$9C5A	CDDEBB	CALL \$BBDE	sélection du stylo graphique

\$9C5D	E1	POP HL	
\$9C5E	010800	LD BC,\$0008	
\$9C61	09	AD HL,BC	
\$9C62	E5	PUSH HL	
\$9C63	ED5BA49C	LD DE,(\$9CA4)	
\$9C67	19	ADD HL,DE	
\$9C68	CD859C	CALL \$9C85	appel du ss-programme de tracé
\$9C6B	3E01	LD A,\$01	
\$9C6D	CDDEBB	CALL \$BBDE	on change de stylo
\$9C70	E1	POP HL	
\$9C71	E5	PUSH HL	
\$9C72	010800	LD BC,\$0008	
\$9C75	CD859C	CALL \$9C85	appel du ss-programme de tracé
\$9C78	E1	POP HL	
\$9C79	E5	PUSH HL	
\$9C7A	ED5BA69C	LD DE,(\$9CA6)	
\$9C7E	A7	AND A	
\$9C7F	ED52	SBC HL,DE	décrémentation nb. de pts
\$9C81	20D5	JR NZ,\$9C58	retour en \$9C58 si non terminé
\$9C83	E1	POP HL	
\$9C84	C9	RET	retour au programme Basic
\$9C85	1160A0	LD DE,\$A060	début du ss-programme de tracé
\$9C88	EDB0	LDIR	
\$9C8A	ED5B60A0	LD DE,(\$A060)	coord x
\$9C8E	2A62A0	LD HL,(\$A062)	coord y
\$9C91	CDC0BB	CALL \$BBC0	appel de la routine MOVE
\$9C94	ED5B64A0	LD DE,(\$A064)	coord x
\$9C98	2A66A0	LD HL,(\$A066)	coord y
\$9C9B	CDF6BB	CALL \$BBF6	appel de la routine DRAW
\$9C9E	C9	RET	